The Spread Of Sentiments Through Social Systems

New Mexico
Super Computing Challenge
April 28, 2008

Team 11
Artesia High School

Team Members:

Tracey Franklin
Jessica MacKinnon
And
Kandese Spikes

Teacher:

Mr. Gaylor

Mentor:

Nick Bennett

Contents

Executive Summary

This project focuses on the effect of social networks on the spread and distribution of ideas in society. Using various principles of Sociology, the hypothesis became the theory that the ideas would eventually reach an equilibrium with one idea potentially dominating the rest. The program mainly displays the separation of people into their social networks. All the while, they are conversing and exchanging ideas based on traits such as charisma and tenacity. After numerous test runs, results indicated that it is highly unlikely that there would be more ideas than the number of networks. Furthermore, the ideas eventually reached a stabilization, proving and adding to the main hypothesis.

Introduction

The definition of Sociology is the study of human social behavior, especially the study of the origins, organization, institutions, and development of human society. Also involved in the definition is that it is analysis of a social institution or societal segment as a self-contained entity or in relation to society as a whole. The purpose of this project is to model how the above mentioned factors inhibit or encourage the spread of ideas. By studying the behaviors of humans, and their everyday living conditions a person might come to the conclusion that depending on how much tenacity or gullibility an individual has determines how they will feel about an idea or a technical innovation, and how likely they are to spread it on to others. These concepts were molded from the sociological idea of "commitment" (Parkinson). Another major factor to consider is the affect of social groups and social networks on the spread of ideas. A simple model which includes human behavior and "clique" behavior can be engineered to fit many different situations. This could help to show viral marketing, product distribution, the effects of the media, conspiracy theories, the spread of religion, etc. The model attempts to solve these problems by determining how an idea or thought spreads through a controlled environment and affects all of its members.

Using the concepts of tenacity, gullibility, charisma, and the basic behavior of cliques as the basis of comparison, this model will show how these factors determine the spread of an idea, or thought. What the concepts really mean is that people who are more gullible are very susceptible and easily accept ideas, people who have high tenacity think over the idea and question it heavily before either accepting or denying it. The Online Dictionary of Social

Sciences used the term “commitment” which means “the degree to which an individual pursues conventional goals.” Thus the agents of the program either have a high commitment or a low commitment. The program defines these as the terms tenacity (high commitment), and gullibility (low commitment). People who have high charisma are more likeable and have an exceptional ability to convince others of their perceptive. This idea of charisma was born of two different sociological ideas: charismatic authority and coercion. The Online Dictionary of Social Sciences defined charismatic authority:

> “The capacity of an individual or institution to secure compliance from others based on the possession of a recognized right to legitimately claim obedience. Authority is obeyed because the individual or institution issuing commands is believed to have the right to do so [furthermore] charismatic [authority is] founded on a belief in a leader's exceptional qualities and inspirational mission; and rational-legal, founded on democratic principles and a framework of law to which all individuals and institutions are subject.” (Parkinson)

The same online dictionary defined coercion as: “[T]he use of force or commands to gain obedience without willing consent of the individual” (Parkinson). Thus some individuals can influence the people they meet whereas others simply cannot. The term “clique,” in this case, can be described by The Online Dictionary of Social Sciences’ term “cohort,” which means: “people sharing a similar experience or event at a particular time” (Parkinson). These shared experiences were created by creating networks of people. Networks of people have been included with the idea of “flocking.” Individuals are given a number, which corresponds to the number of the clique in which they belong, and three instructions that give them the need to travel in groups of their same number. The RiverSoftAVG website calls these three instructions “separation, alignment, and cohesion” (Flocking). Patterns appear once the individuals are told to keep a minimal distance from others of their number (separation), to try to go the same direction of the

others of their number (alignment), and to try huddle with others of their number (cohesion).

The level of tenacity or gullibility, and the level of charisma for each individual are determined by “normal distribution” (more commonly called a bell curve) because “it is a useful tool for describing many random phenomena [such as] measurements of many human activities” (Tabak 42). The bell curve “has a single central peak and tapers off on either side in a symmetrical fashion” (Gibilisco 87). The program’s central point is fifty on a graph from one to one hundred. Its standard deviation is twenty. (see appendix A) This means that, while some agents fall elsewhere, the majority of them have charismas or tenacities between thirty and seventy, most falling closer to fifty. The model has the ability to perform multiple trials which is not possible in reality and therefore can give solutions to the problem not feasible to acquire in real life.

Description of Research

In order for the program to solve these problems our turtles have to accurately mimic actual human behavior. Much research has gone into this project in order to accomplish this. The project’s research has involved both an in-depth look at the human mind, and human behavior, as

well as a look into how different types of media, new innovations, ideas, and rumors manage to spread throughout society. These many aspects have been proven to be linked to human sociology. The beginning stages of the project involved many assumptions that were made upon personal observations. These assumptions were then tested. This happened by encyclopedia, book, and online research which was meant to confirm that others had noticed the same behaviors. The reverse also took place. At times new ideas were found from an outside source and then observation was needed to confirm these ideas. Sociological dictionaries were particularly helpful in all phases of the project

Model

The model itself is run through an agent-based program called NetLogo. First, a number of agents, determined from a set of sliders, are created with an idea numbered zero through four (also determined from a slider). These agents are then assigned a charisma and a tenacity based on the bell curve mentioned above. Next, they are assigned a “clique,” or social network. The maximum number of “cliques” is also determined by a slider. The running of the model is

basically composed of four steps: flock, move, duel, and update turtles.

The first step, flocking, follows the rules of separation, alignment, and cohesion as mentioned before. This makes the agents turn themselves toward other members of their cliques. Eventually the agents will join together into separate groups based on their clique number and move as such. Each of these groups, or "cliques" according to the program, represents a social network.

To move, each turtle turns what is called a "random float" of ten, which means any number, including decimals, between zero and ten. Next, they move forward a float of one (any number between zero and one).

The procedure "duel" is the step in which agents actually exchange ideas. Basically, an agent picks a partner, preferably a member of his own clique. Next, they compare ideas. If the ideas are different, partners then compare charismas and tenacities. If the agent's tenacity is lower than its partner's charisma, it then subtracts the difference between the two from a temporary attribute called "tempTenacity." If the ideas of the partners are the same, the difference between the two tenacities is added to the agent's tempTenacity.

The final procedure, update-turtles, changes agents after they decide to convert ideas (when tempTenacity drops below zero). Using an array, the agent determines which idea has had the most influence and the greatest charisma. She/he will then convert to that idea.

For code, see appendix B.

History

Originally the program was meant to be “The Spread and Advancement of Technology in Society,” however it was quickly reduced to a simpler project. Mainly, the project aimed to model the spread of ideas in society. The first attempt centered on rumors. However, when flocking was added to the program to incorporate cliques, the entire project changed to what it is now: The Effect of Social Networks on the Spread of Ideas.

Tests and Results

NetLogo comes equipped with an attachment called Behavior Space. This attachment runs the program numerous times and takes specific results. The experiments run on our program included varying the number of social networks. For detailed, numeric results, see appendix C.

With these results, it can be determined that the maximum number of ideas is equal to the number of social networks. For example, if there is only one social network, one idea would dominate. If there are two, two ideas will become approximately equal while the others die out, or one may dominate. With five cliques, the five ideas may spread equally, or a number may unequally prosper while the others die out. However, with ten networks and only five ideas, the

results become more random and unpredictable. It becomes what is called a “Monte Carlo model.” A Monte Carlo model is a program that is different every run. No definite patterns can be discovered.

Thus, it becomes necessary to watch the model. Results from these views found that generally a social network will be dominated by one idea and remain that idea for the remainder of the test. This explains the above discovery. It also describes the randomness of the final test with ten networks. As the model progresses, the different ideas spread out among the networks. This distribution is different every time, and thus far a pattern has not been discovered.

Conclusion

The completed program has become an ever evolving prototype from the first trial run through our continued tests. The test runs have produced patterns and trends. These patterns have both proven and refuted our hypothesis and continue to teach about the effects of personalities, groups, and idea potential on the infectious-spread rate of ideas. These tests, when compared with premeditated research, seem to closely mirror true life. Just as predicted, the ideas eventually reached a stable balance, just as many things in life become balanced excepting possible extraneous circumstances.

Bibliography

"Flocking." RiverSoftAVG. 01 Feb. 2008. RiverSoftAVG. 18 Mar. 2008 <http://ww.riversoftavg.com/flocking.htm>.

Gibilisco, Stan. Statistics Demystified: a Self-Teaching Guide. New York, NY: McGraHill Companies, Inc., 2004. 87-88.

Parkinson, Gary, PhD, and Drislane, Robert, PhD. "The Online Dictionary of Social Sciences. "Bitbucket.Icaap. ICAAP for Athabasca University. 18 Mar. 2008 <http://bitbucket.icaap.org/dict.pl?alpha=A>.

Tabak, John. Probability and Statistics. New York, NY: Facts on File, Inc., 2004. 42 – 43.

Appendix A


50
0
100
©Copyright, Robert Niles, http://www.robertniles.com/stats/stdev.shtml

Appendix B

```
;;The array is used to keep track of which idea has influenced an agent more
extensions [array]


turtles-own [
    idea                                    ;;which agent has what idea
    charisma                ;;used in communication
    tenacity                ;;a stored, permanent trait
    tempTenacity            ;;what changes with influence
    clique                  ;;or social network
    partner                 ;;used during exchanging of ideas
    flockmates              ;;used in flocking
    nearest-neighbor                ;;used in flocking
    charisma-encountered    ;;this is the array previously mentioned
```

```
]


;;This procedure 'sets up the board' for play.
;;It clears the board, creates a number of agents with certain ideas based on sliders.
;;Then it sets up all the agents' attributes.
to setup
  clear-all
  create-turtles idea-1 [
    set color pink
    set idea 0
  ]
  create-turtles idea-2 [
    set color green
    set idea 1
  ]
  create-turtles idea-3 [
    set color blue
    set idea 2
  ]
  create-turtles idea-4 [
    set color yellow
    set idea 3
  ]
```

```
  create-turtles idea-5 [
    set color red
    set idea 4
  ]
  ask turtles [
    setxy random-xcor random-ycor
    set charisma max (list .1 (random-normal 50 20))
    set tenacity max (list .1 (random-normal 50 20))
    set tempTenacity tenacity
    set clique random total-cliques
    set partner nobody
    set charisma-encountered array:from-list n-values 5 [0]
  ]
end

;;This is the run procedure which puts all procedures together.
to go
  ask turtles [
    flock
    move
    duel
    update-turtles
  ]
```

```
    tick
    update-plot
end

;;This is basic random movement.
to move
    right ((random-float 10) - (random-float 10))
    forward (random-float 1)
end

;;This is where exchanging of ideas occurs
;;An agent picks a partner, preferably someone in his network
;;Then, if the ideas are different, they compare tempTenacities and Charismas.
;;If tempTenacity is lower than the partner's charisma, they subtract the difference from their
;;tempTenacity and add the charisma to the particular basket for that partner's idea.
;;If ideas are the same, they increase tempTenacity by the difference of the tempTenacities.
to duel
    if any? other turtles in-radius 1 [
        if (partner = nobody) [
            let peers (other turtles in-radius 1 with [partner = nobody])
            if (any? peers) [
                let clique-peers (peers with [clique = [clique] of myself])
                ifelse (any? clique-peers) [
```

```
        set partner (one-of clique-peers)
    ] [
        set partner (one-of peers)
    ]
    ask partner [
        set partner myself
    ]
]
if partner != nobody and [idea] of partner != idea [
    if tempTenacity < [charisma] of partner [
        set tempTenacity (tempTenacity - ([charisma] of partner - tempTenacity))
        array:set charisma-encountered ([idea] of partner)
        (array:item charisma-encountered ([idea] of partner) + ([charisma] of partner))
    ]
    if charisma > [tempTenacity] of partner [
        ask partner [
            set tempTenacity (tempTenacity - ([charisma] of myself - tempTenacity))
            array:set charisma-encountered ([idea] of myself)
            (array:item charisma-encountered ([idea] of myself) + ([charisma] of myself))
        ]
    ]
]
if partner != nobody and [idea] of partner = idea [
```

```
            set tempTenacity tempTenacity + (abs ([tempTenacity] of partner - tempTenacity))
        ]
    ]
  ]
end

;;This is where agents change colors and ideas.
;;If an agent's tempTenacity falls below 0, it looks at its array.
;;It sees which item in the array is greater, then converts to that idea.
to update-turtles
  set tempTenacity (min list tenacity tempTenacity)
  if tempTenacity <= 0 [
    let new-idea -1
    let new-idea-charisma -1
    foreach (n-values 5 [?]) [
      let test-charisma (array:item charisma-encountered ?)
      if test-charisma > new-idea-charisma [
        set new-idea-charisma test-charisma
        set new-idea ?
      ]
    ]
    set idea new-idea
    set charisma-encountered array:from-list n-values 5 [0]
```

```
        set tempTenacity tenacity
        if idea = 0 [
            set color pink
        ]
        if idea = 1 [
            set color green
        ]
        if idea = 2 [
            set color blue
        ]
        if idea = 3 [
            set color yellow
        ]
        if idea = 4 [
            set color red
        ]
    ]
    set partner nobody
end

;;Flocking is the idea that creatures stick together.
;;This is what keeps the turtles together, following each other.
to flock
```

```
  find-flockmates
  if any? flockmates [
    find-nearest-neighbor
    ifelse distance nearest-neighbor < minimum-separation [
      separate
    ] [
      align
      cohere
    ]
  ]
end

;;Flocking procedure
to find-flockmates
  let my-clique clique
  set flockmates (other ((turtles in-radius vision) with [clique = my-clique]))
end

;;Flocking procedure
to find-nearest-neighbor
  set nearest-neighbor min-one-of flockmates [distance myself]
end
```

```
;;Flocking procedure
to separate
  turn-away ([heading] of nearest-neighbor) max-separate-turn
end
;;Flocking procedure
to align
  turn-towards average-flockmate-heading max-align-turn
end

;;Flocking procedure
to-report average-flockmate-heading
  report atan sum [sin heading] of flockmates
                  sum [cos heading] of flockmates
end

;;Flocking procedure
to cohere
  turn-towards average-heading-towards-flockmates max-cohere-turn
end

;;Flocking procedure
to-report average-heading-towards-flockmates
  report atan mean [sin (towards myself + 180)] of flockmates
```

```
                mean [cos (towards myself + 180)] of flockmates
end

;;Flocking procedure
to turn-towards [new-heading max-turn]
   turn-at-most (subtract-headings new-heading heading) max-turn
end

;;Flocking procedure
to turn-away [new-heading max-turn]
   turn-at-most (subtract-headings heading new-heading) max-turn
end

;;Flocking procedure
to turn-at-most [turn max-turn]
   ifelse abs turn > max-turn
      [ ifelse turn > 0
            [ rt max-turn ]
            [ lt max-turn ] ]
      [ rt turn ]
end

;;This procedure updates the graph.
```

```
to update-plot
    set-current-plot "Total Population"
    set-current-plot-pen "idea-1"
    plot count turtles with [idea = 0]
    set-current-plot-pen "idea-2"
    plot count turtles with [idea = 1]
    set-current-plot-pen "idea-3"
    plot count turtles with [idea = 2]
    set-current-plot-pen "idea-4"
    plot count turtles with [idea = 3]
    set-current-plot-pen "idea-5"
    plot count turtles with [idea = 4]
end
```

Appendix C

Charts

No Social Networks

	Idea 1	Idea 2	Idea 3	Idea 4	Idea 5
Run 1	0	0	0	0	250
Run 2	0	0	250	0	0
Run 3	0	1	161	4	84
Run 4	0	250	0	0	0
Run 5	0	0	250	0	0
Run 6	249	1	0	0	0
Run 7	2	0	0	248	0
Run 8	0	249	0	1	0
Run 9	2	1	2	0	245
Run 10	0	0	249	0	1

2 Social Networks

	Idea 1	Idea 2	Idea 3	Idea 4	Idea 5
Run 1	1	0	135	113	1
Run 2	3	1	1	0	245
Run 3	0	118	2	130	0
Run 4	113	0	137	0	0
Run 5	5	0	114	131	0
Run 6	248	0	0	1	1
Run 7	1	3	138	108	0
Run 8	122	57	0	1	70
Run 9	0	0	129	0	121
Run 10	0	0	129	121	0

5 Social Networks

	Idea 1	Idea 2	Idea 3	Idea 4	Idea 5
Run 1	47	46	53	66	38
Run 2	93	3	53	98	3
Run 3	1	103	99	46	1
Run 4	8	52	0	187	3
Run 5	52	46	2	4	146
Run 6	36	165	4	45	0
Run 7	0	94	153	1	2
Run 8	104	57	50	36	3
Run 9	5	114	52	45	34

Run 10	50	1	52	106	41

10 Social Networks

	Idea 1	**Idea 2**	**Idea 3**	**Idea 4**	**Idea 5**
Run 1	0	185	0	60	5
Run 2	0	0	43	77	130
Run 3	94	31	84	21	20
Run 4	61	26	34	103	26
Run 5	26	0	130	34	60
Run 6	41	149	4	56	0
Run 7	52	147	28	23	0
Run 8	96	20	30	103	1
Run 9	27	38	59	78	48
Run 10	153	43	2	0	52